HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Heating, Ventilating & Air Conditioning



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁎. When you see ⁎ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^T SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^T THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↑*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^T R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

* These programs require one memory module.
** This program requires two memory modules.

OVERALL HEAT TRANSFER COEFFICIENTS

This program sums convective and conductive coefficients to obtain the overall heat transfer coefficient for walls.

Equations:

$$U = \frac{1}{R_0 + R_1 + R_2 + \ldots + R_n + \frac{1}{h_0} + \frac{1}{h_1} + \frac{1}{h_2} + \ldots + \frac{1}{h_n} + \frac{x}{k_0} + \frac{x}{k_1} + \frac{x}{k_2} + \ldots + \frac{x}{k_n} + \frac{1}{C_0} + \frac{1}{C_1} + \frac{1}{C_2} + \ldots + \frac{1}{C_n}}$$

where:

- Q is the heat transfer
- U is the overall heat transfer coefficient
- ΔT is the temperature difference across the wall
- R is the thermal resistance of a layer in the wall
- h is the convective resistance at a surface of the wall
- x is the thickness of a conductive layer
- k is the thermal conductivity of a substance
- C is the thermal conductance of a layer

Reference:

Threlkeld, James L.; Thermal Environmental Engineering; Prentice-Hall, Inc. 1970

Example:

What is the overall heat transfer coefficient for the wall below? What is the heat transfer for a 20°F temperature difference?



Layer	Coefficient	Symbol	Units
Inside Air	1.46	h	B/(hr) (ft^2) (°F)
Plaster	5	k	B/(hr) (ft^2) (°F/in)
Insulation	11	R	(hr) (ft^2) (°F)/BTU
Concrete	12	h	BTU/(hr) (ft^2) (°F/in)
Brick	36	C	BTU/(hr) (ft^2) (°F)
Outside Air	6	h	BTU/(hr) (ft^2) (°F)

Keystrokes:	Display:
[///] [FIX] 4	
[XEQ] [ALPHA] HEAT [ALPHA]	H X↑K C R CL
1.46 [A]	H X↑K C R CL
.5 [ENTER ↑] 5 [B]	H X↑K C R CL
11 [D]	H X↑K C R CL
5 [ENTER ↑] 12 [B]	H X↑K C R CL
36 [C]	H X↑K C R CL
6 [A]	H X↑K C R CL
[R/S]	U=0.0807
20 [x]	1.6134 (Btu/(ft^2)(hr))

User Instructions

SIZE: ANY

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Initialize program		[XEQ] HEAT	H X↑K C R CL
3.	Note that the top row of keys are now			
	defined by the display.			
4a.	To add a convective coefficient key it in			
	and press [A]	h	[A]	H X↑K C R CL
4b.	To add a conductive coefficient key in the			
	thickness and press [ENTER↑],	x	[ENTER]	X
	then the coefficient and press [B].	k	[B]	H X↑K C R CL
4c.	To add a conductance key it in and press			
	[C]	C	[C]	H X↑K C R CL
4d.	To add a resistance key it in and press			
	[D]	R	[D]	H X↑K C R CL
4e.	To clear the sum press [E]		[E]	H X↑K C R CL
5.	Return to step 4 for all coefficients.			
6.	Display overall coefficient		[R/S]	U =
7.	Key in temperature difference and compute			
	heat transfer	ΔT	[x]	Q
8.	For a new case go to step 4e.			

Program Listings

Program steps	Comments
01♦LBL "HEAT" 02♦LBL E 03 ADV 04 SF 27 05 CLST	Initialize program.
06♦LBL 01 07 "H X↑K CR CL" 08 PROMPT 09 ADV	Define keyboard.
10 "U=" 11 ARCL X 12 SF 21 13 AVIEW 14 GTO 01	Output U.
15♦LBL A 16 "H" 17 XEQ 00 18 1/X 19 GTO 02	Add convective coefficient.
20♦LBL B 21 "X=" 22 ARCL Y 23 FS? 55 24 PRA 25 "K" 26 XEQ 00 27 / 28 GTO 02	Add conductive coefficient.
29♦LBL C 30 "C" 31 XEQ 00 32 1/X 33 GTO 02	Add coductance.
34♦LBL D 35 "R" 36 XEQ 00 37♦LBL 02 38 RCL Y 39 X≠0? 40 1/X 41 + 42 1/X	Add resistance.
43 GTO 01 44♦LBL 00 45 "⊢=" 46 ARCL X 47 FS? 55 48 PRA 49 RTN	Print input.

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE ANY TOT. REG. 15 USER MODE
ENG ____ FIX ____ SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

INSULATION BREAK EVEN ANALYSIS

This program calculates the number of years necessary for insulation to pay for itself based on heating costs. Inflation and the cost of money are accounted for.

Equations:

$$\text{YEARS} = \frac{-\ln\left(1 - \dfrac{\$/\text{FT}^2\,(\%\text{INT}-\%\text{INF})}{\text{Yearly Savings}}\right)}{\ln\left(1 + \dfrac{\%\text{INT}-\%\text{INF}}{1 + \%\text{INF}}\right)}$$

$$\text{YEARLY SAVINGS} = \left(U - \frac{1}{(1/U + R)}\right)(\text{DEGDY})\,(24)\,(\$/\text{BTU})$$

where:

YEARS is the number of years before insulation pays for itself.

$\$/FT^2$ is the cost of the insulation on a square foot basis.

%INT is the current lending rate to obtain insulation.

%INF is the expected inflation rate for the heat source being considered.

U is the overall heat transfer coefficient for the surface being considered with no insulation installed, Btu/(hr)(ft^2)(°F).

R is the thermal resistance of the insulating material, (hr)(ft^2)(°F)/BTU.

DEGDY is the number of degree days for the area being considered. See the ASHRAE GUIDE AND DATA BOOK for degree day information.

\$/BTU is the current cost of heat per Btu.

Example:

A home in Seattle, Washington, (5145 degree days per year) is to have walls insulated with either R11 or R19 insulation costing 0.25 or 0.40 dollars per square foot respectively. The heat transfer coefficient for the un-insulated walls is 0.28 Btu/(hr)(ft^2)(°F). Natural gas at a cost of 3.66×10^{-6} \$/Btu is the heat source. Financing is available at 11% and energy is expected to inflate at 15% per year. How many years will it take for the two insulation choices to pay for themselves?

Keystrokes: (SIZE ⩾ 009)	Display:
[///] [FIX] 2	
[XEQ] [ALPHA] INSUL [ALPHA]	U=?
.28 [R/S]	$/BTU=?
3.66 [EEX] [CHS] 6 [R/S]	% INT=?
11 [R/S]	% INF=?
15 [R/S]	DEGDY=?
5145 [R/S]	R=?
11 [R/S]	$/FT2=?
.25 [R/S]	YEARS=2.81
[R/S]*	R=?
19 [R/S]	$/FT2
.4 [R/S]	YEARS=3.95
[R/S]*	R=?

*Press [R/S] if you are not using a printer.

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Initialize program		[XEQ] INSUL	U = ?
3.	Key in overall heat transfer coefficient			
	for uninsulated wall, floor, or ceiling	U	[R/S]	$/BTU=?
4.	Key in per BTU cost of heat	$/BTU	[R/S]	%INT=?
5.	Key in interest rate	%INT	[R/S]	%INF=?
6.	Key in expected fuel inflation rate	%INF	[R/S]	DEGDY=?
7.	Key in degree days	DEGDY	[R/S]	R=?
8.	Key in insulation R value	R	[R/S]	$/FT2=?
9.	Key in cost of insulation per square foot			
	of insulated surface	$/FT2	[R/S]	YEARS=
			[R/S]*	R=?
10.	For a new R value and/or cost go to step 8.			
	For a new case go to step 2. For values			
	which do not change, you may just press			
	[R/S].			
*	Press [R/S] if you are not using a printer.			

Program Listings

Input values.

```
01♦LBL "INS
UL"
02 ADV
03 0
04 STO 00
05 "U"
06 XEQ "IN"
07 "$/BTU"
08 XEQ "IN"
09 "%INT"
10 XEQ "IN"
11 "%INF"
12 XEQ "IN"
13 "DEGDY"
14 XEQ "IN"
15♦LBL 01
16 5
17 STO 00
18 "R"
19 XEQ "IN"
20 "$/FT2"
21 XEQ "IN"
```

Calculate savings.

```
22 RCL 01
23 RCL 01
24 1/X
25 RCL 06
26 +
27 1/X
28 -
29 RCL 05
30 *
31 RCL 02
32 *
33 24
34 *
```

Calculate years.

```
35 RCL 03
36 RCL 04
37 -
38 X=0?
39 GTO 00
40 100
41 /
42 STO 08
43 /
44 RCL 07
45 X<>Y
46 /
47 CHS
48 LN1+X
49 CHS
50 RCL 08
51 RCL 04
52 100
53 /
54 1
55 +
56 /
57 LN1+X
58 /
```

Output years.

```
59♦LBL 03
60 ADV
61 "YEARS"
62 XEQ "O"
63 ADV
64 GTO 01
```

Special case where inflation equals interest.

```
65♦LBL 00
66 RDN
67 RCL 07
68 RCL 04
69 %
70 +
71 X<>Y
72 /
73 GTO 03
```

Input subroutine.

```
74♦LBL "IN"
75 CF 22
76 1
77 ST+ 00
78 RCL IND
00
79 "⊢="
80 ASTO Y
81 "⊢?"
82 CF 21
83 AVIEW
84 SF 21
85 CLA
86 ARCL Y
87 STOP
88 STO IND
00
89 FS? 22
90 FC? 55
91 RTN
92 ARCL X
93 PRA
94 RTN
```

Output subroutine

```
95♦LBL "O"
96 "⊢="
97 ARCL X
98 AVIEW
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	DATA PRINTER	50	
	U		
	$/BTU		
	%INT		
	%INF		
05	DEGDY	55	
	R		
	$/FT2		
	(%INT-%INF)/100		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 37 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

AIR FLOW IN CIRCULAR DUCTS

(Requires one memory module)

This program can be used to compute the pressure difference, velocity or volumetric flow rate for air in circular, metal, HVAC ducts.

Equations:

$$v^2 = \frac{dP/\rho}{2\left(f\frac{L}{D} + \frac{K_T}{4}\right)}$$

For laminar flow (Re < 2300)

$$f = 16/Re$$

For turbulent flow (Re > 2300)

$$\frac{1}{\sqrt{f}} = 1.737 \ln \frac{D}{\varepsilon} + 2.28 - 1.737 \ln \left(4.67 \frac{D}{\varepsilon \, Re \sqrt{f}} + 1\right)$$

is solved by Newton's method.

$$\frac{1}{\sqrt{f_0}} = 1.737 \ln \frac{D}{\varepsilon} + 2.28$$

is used as an initial guess in the iteration.

$$\mu = 1.101 \times 10^{-6} \; (T)^{1.5}/(T+200) \quad (N \cdot s/m^2)$$

$$\rho = 1.201 \left(\frac{P}{29.92}\right)\left(\frac{530}{T}\right) \quad (kg/m^3)$$

where:

- Re is the Reynolds number, defined as $\rho Dv/\mu$;
- D is the pipe diameter;
- ε is the dimension of irregularities in the conduit surface
- f is the Fanning friction factor for conduit flow;
- dP is the pressure difference along the conduit;
- ρ is the density of the fluid;
- μ is the viscosity of the fluid;
- L is the conduit length;

v is the average fluid velocity;

K_T is the total of the applicable fitting coefficients in table 1;

T is the air temperature in degrees ranking.

P is the air pressure in inches of mercury.

Table 1

Fitting Coefficients

Fitting	K
90° elbow	0.4-0.9
Standard 45° elbow	0.35-0.42
Tee, through side outlet	1.5
Tee, straight through	.4
180° bend	1.6
Entrance to circular pipe	0.25-0.50
Sudden expansion	$(1 - A_{up}/A_{dn})^2$*
Acceleration from v = 0 to $v = V_{entrance}$	1.0

Remarks:

For rectangular ducts, use the duct conversion program of this solution book.

This program assumes incompressible flow, thus accuracy will degenerate as velocity increases above 12,000 feet per minute.

No algorythm gives reliable outputs for transition flow (2300 < Re < 4000). If this condition is encountered, the program will hault displaying "TRANSITION". You may press [R/S] and the program will continue to an answer but the answer may have little or no relation to reality.

References:

Welty, Wicks, Wilson; *Fundamentals of Momentum, Heat and Mass Transfer*, John Wiley and Sons, Inc., 1969.

Baumeister, *Mark's Standard Handbook For Mechanical Engineers*, Seventh Edition, McGraw-Hill, 1967, p. 12-136.

*A_{up} is the upstream area and A_{dn} is the downstream area.

Example 1:

What pressure drop will force 6000 cubic feet per minute of 70°F air through 100 feet of 20 inch straight conduit. The atmospheric pressure is 29 inches of mercury.

Keystrokes: (SIZE ≥ 018)	Display:
[///] [FIX]	
[XEQ] [ALPHA] AIRDUCT [ALPHA]	T,F=?
70 [R/S]	INHG=?
29 [R/S]	L,FT=?
100 [R/S]	ΣK=?
0 [R/S]	D,IN=?
20 [R/S]	SELECT KEY: dP V Q
[C]	Q,CFM=?
6000 [R/S]	INH20=0.41
[R/S]*	V,FPM=2750.73
[R/S]*	D,IN=?

* Press [R/S] if you are not using a printer.

Example 2:

The flow rate in Example 1 is to be increased to 15,000 cubic feet per minute by increasing the conduit diameter. What diameter is necessary?

Keystrokes: Display:

First run Example 1. You should have the "D,IN=?" prompt in the Display. Guess a diameter of 30 inches.

Keystrokes	Display
30 [R/S]	SELECT KEY: dP V Q
[A]	INH20=?
[R/S]	V,FPM=3,551.69
[R/S]*	Q,CFM=17,430.93
[R/S]*	D,IN=?

The 30 inch duct will be adequate but try a 28 inch duct.

Keystrokes	Display
28 [R/S]	SELECT KEY: dP V Q
[A] [R/S]	V,FPM=3,401.48
[R/S]*	Q,CRM=14,542.07
[R/S]*	D,IN=?

Since duct is available only in even size use 30 inch duct.

User Instructions

SIZE: 018

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Initialize program		[XEQ] AIRDUCT	T,F=?
3.	Key in air temperature in degrees			
	Fahrenheit.	T	[R/S]	INHG=?
4.	Key in average atmospheric pressure in			
	inches of mercury.	P_{atm}	[R/S]	L,FT=?
5.	Key in length of duct in feet	L	[R/S]	ΣK
6.	Key in sum of fitting coefficients from			
	Table 1.	ΣK	[R/S]	D,IN=?
7.	Key in duct diameter in inches	D	[R/S]	SELECT KEY:
				dP V Q
8.	If you know pressure difference, press			
	[A]		[A]	INH2O=?
	If you know velocity, press [B]		[B]	V,FPM=?
	If you know volumetric flow rate press			
	[C]		[C]	Q, CFM=?
9.	Key in known selected in step 8 and			
	calculate remaining unknowns.	known	[R/S]	$UNKNOWN_1$
			[R/S]*	$UNKNOWN_2$
			[R/S]*	D,IN=?
10.	For a new diameter, or step 8 input, go to			
	step 7. For a new case go to step 2. In			
	either case, key in only values which			
	change.			
*	Press [R/S] if you are not using a printer.			

Program Listings

01♦LBL "AIR DUCT" 02 ADV 03 0 04 STO 00 05 "T,F" 06 XEQ "IN" 07 "INHG" 08 XEQ "IN" 09 RCL 01 10 460 11 + 12 STO 10 13 RCL X 14 1.5 15 Y↑X 16 1.101 E-6 17 * 18 RCL Y 19 200 20 + 21 / 22 STO 16 23 RCL 02 24 29.92 25 / 26 530 27 RCL 10 28 / 29 * 30 1.201 31 * 32 STO 17 33 762 E-7 34 STO 03	Calculate and store viscosity, density and surface irregularities.
35 3 36 STO 00 37 "L,FT" 38 .3048 39 ST/ 04 40 XEQ "IN" 41 .3048 42 ST* 04 43 1	Input L.
44 ST+ 00 45 "ΣK" 46 XEQ "IN"	Input sum of fitting coefficients.
47♦LBL 15 48 4 49 STO 00	
50 "D,IN" 51 .0254 52 ST/ 05 53 XEQ "IN" 54 .0254	Input D.
55 ST* 05 56 SF 27 57 "SELECT KEY:" 58 CF 21 59 AVIEW 60 SF 21 61 PSE 62♦LBL 01 63 "dP V Q" 64 PROMPT 65 GTO 01	Define keys.
66♦LBL A 67 6 68 STO 00 69 248.8 70 ST/ 07 71 "INH2O" 72 XEQ "IN" 73 248.8 74 ST* 07 75 SF 03 76 XEQ 09	Input pressure difference.
77♦LBL 03 78 RND 79 STO 15 80 XEQ 08 81 RND 82 RCL 15 83 X<>Y 84 X≠Y? 85 GTO 03 86 XEQ "Re" 87 "V,FPM" 88 RCL 08 89 196.9 90 * 91 GTO 00	Compute velocity.
92♦LBL C 93 7 94 STO 00 95 "Q,CFM" 96 XEQ 11 97 * 98 * 99 STO 08	Input flow rate.

Program Listings

```
100 XEQ "IN"
101 XEQ 11
102 *
103 /
```

Calculate pressure drop.

```
104 STO 08
105 XEQ 10
106 XEQ "Re"
107 "INH2O"
108 RCL 07
109 248.8
110 /
111 XEQ "O"
112 "V,FPM"
113 RCL 08
114 196.9
115 *
116 XEQ "O"
117 ADV
```

Input velocity.

```
118 GTO 15
119◆LBL B
120 7
121 STO 00
122 "V,FPM"
123 5.08 E-3
124 ST/ 08
125 XEQ "IN"
126 5.08 E-3
```

Calculate pressure drop.

```
127 ST* 08
128 XEQ 10
```

Output routine.

```
129 XEQ "Re"
130 "INH2O"
131 RCL 07
132 248.8
133 /
134◆LBL 00
135 XEQ "O"
136 "Q,CFM"
137 XEQ 11
138 *
139 *
140 XEQ "O"
141 ADV
142 GTO 15
```

Calculate pressure drop.

```
143◆LBL 10
144 CF 03
145 XEQ 09
146 RCL 08
147 X↑2
148 *
149 RCL 17
150 *
151 STO 07
152 RTN
```

Guess $1/\sqrt{f}$.

```
153◆LBL 09
154 ADV
155 RCL 16
156 RCL 17
157 /
158 STO 09
159 RCL 05
160 RCL 03
161 /
162 STO 10
163 LN
164 1.737
165 STO 11
166 *
167 2.28
168 +
169 STO 12
170 STO 13
```

Special case for low velocity flow.

```
171 FS? 03
172 GTO 07
173◆LBL 08
174 16
175 RCL 08
176 RCL 05
177 *
178 RCL 09
179 /
180 STO 14
181 2300
182 X<=Y?
183 GTO 02
184 RDN
185 /
186 SQRT
187 1/X
188 STO 13
189 GTO 07
```

Turbulent flow friction factor.

```
190◆LBL 02
191 RCL 12
192 RCL 13
193 -
194 4.67
195 RCL 10
196 *
197 RCL 14
198 /
199 RCL 13
200 *
201 1
```

Program Listings

```
202 +
203 STO 00
204 LN
205 RCL 11
206 *
207 -
208 RCL 00
209 1/X
210 CHS
211 1
212 +
213 RCL 11
214 *
215 RCL 13
216 /
217 1
218 +
219 /
220 ST+ 13
221 RCL 13
222 /
223 ABS
224 1 E-3
225 X<=Y?
226 GTO 02
```

Compute velocity.

```
227♦LBL 07
228 RCL 13
229 1/X
230 X↑2
231 RCL 04
232 *
233 RCL 05
234 /
235 RCL 06
236 4
237 /
238 +
239 ST+ X
240 RCL 07
241 RCL 17
242 /
243 X<>Y
244 FC? 03
245 RTN
246 /
247 SQRT
248 STO 08
249 RTN
```

Check for transition.

```
250♦LBL "Re"
251 "TRANSIT
ION"
252 2300
253 RCL 14
254 X<=Y?
255 RTN
256 4000
257 X>Y?
258 PROMPT
259 RTN
```

Convert between velocity and volumetric flow rate.

```
260♦LBL 11
261 RCL 08
262 RCL 05
263 X↑2
264 PI
265 *
266 4
267 /
268 2119
269 RTN
```

Input routine.

```
270♦LBL "IN"
271 CF 22
272 1
273 ST+ 00
274 RCL IND
00
275 "⊢="
276 ASTO Y
277 "⊢?"
278 CF 21
279 AVIEW
280 SF 21
281 CLA
282 ARCL Y
283 STOP
284 STO IND
00
285 FS? 22
286 FC? 55
287 RTN
288 ARCL X
289 PRA
290 RTN
```

Output routine.

```
291♦LBL "O"
292 "⊢="
293 ARCL X
294 AVIEW
```

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Data Pointer	50	
	T		
	P		
	ε		
	L		
05	D	55	
	ΣK		
	dP		
	V/Q		
	μ/ρ		
10	D/ε	60	
	1.737		
	$1/\sqrt{f}$		
	$1/\sqrt{f}$		
	Re		
15	V	65	
	μ		
	δ		
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 018 TOT. REG. 107 USER MODE
ENG ____ FIX ____ SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
3		V calculation	dP calculation
27		USER mode	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

AIR DUCT CONVERSION

This program converts a round, metal, HVAC, air duct to an equivalent rectangular air duct. The inverse conversion is also provided.

Equations:

$$D = \frac{1.3\ (ab)^{0.625}}{(a + b)^{0.25}}$$

where:

a is the depth (or width) of a rectangular duct in inches.

b is the width (or depth) of a rectangular duct in inches.

D is the approximate diameter for an equivalent round duct.

Reference:

Wolfe, James M.; Air Duct Conversion; HP-67/97 Users' Library Program 03046D.
Baumeister, Mark's Standard Handbook for Mechanical Engineers, Seventh Edition, McGraw Hill, 1967, p. 12-137

Example 1:

A duct measuring 15 inches by 8 inches is to be replaced with a round duct. What diameter duct is required?

Keystrokes: (SIZE $\geq$ 003)	Display:
[///] [FIX] 2	
[XEQ] [ALPHA] REC-RND [ALPHA]	a=?
15 [R/S]	b=?
8 [R/S]	D=12.00
[R/S]*	a=?

Example 2:

A duct sizing program computes a diameter of 30 inches for a duct. Clearances only allow 24 inches in the vertical direction. What horizontal dimension is necessary for a rectangular duct?

Keystrokes: (SIZE $\geq$ 003)	Display:
[///] [FIX] 2	
[XEQ] [ALPHA] RND-REC [ALPHA]	D=?
30 [R/S]	a=?
24 [R/S]	b=32
[R/S]*	D=?

*Press [R/S] if you are not using a printer.

User Instructions

SIZE: 003

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	For circular to rectangular conversions,			
	go to step 3b.			
3.	Initialize for rectangular to circular			
	conversion.		[XEQ] REC-RND	a=?
4.	Key in depth.	a	[R/S]	b=?
5.	Key in width and calculate diameter.	b	[R/S]	D=
			[R/S]*	a=?
6.	For a new case go to step 4a. Key in			
	only values which change. For the			
	inverse conversion, go to step 3b.			
3b.	Initialize for circular to rectangular			
	conversion		[XEQ] RND-REC	D=?
4b.	Key in diameter	D	[R/S]	a=?
5b.	Key in the depth or width and calculate			
	the other dimension	a	[R/S]	b=
			[R/S]*	D=?
6b.	For a new case, go to step 4b. Key in			
	only values which change. For the			
	inverse conversion, go to step 3.			
*	Press [R/S] if you are not using a printer.			

Program Listings

Input a and b.

```
01♦LBL "REC
-RND"
02♦LBL 01
03 ADV
04 0
05 STO 00
06 "a"
07 XEQ "IN"
08 "b"
09 XEQ "IN"
10 ADV
```

Calculate D.

```
11 RCL 01
12 *
13 .625
14 Y↑X
15 RCL 01
16 RCL 02
17 +
18 SQRT
19 SQRT
20 /
21 1.3
22 *
23 "D"
24 .5
25 +
26 INT
27 XEQ "O"
28 GTO 01
```

Input D and a.

```
29♦LBL "RND
-REC"
30♦LBL 02
31 ADV
32 0
33 STO 00
34 "D"
35 XEQ "IN"
36 "a"
37 XEQ "IN"
```

Guess b based on wetted perimeter.

```
38 RCL 02
39 RCL 01
40 *
41 RCL 02
42 ST+ X
43 RCL 01
44 -
45 /
46 STO 00
47 ADV
48♦LBL 03
```

Iterate by Newton's method to find b.

```
49 RCL 00
50 RCL 02
51 *
52 .625
53 Y↑X
54 1.3
55 *
56 RCL 00
57 RCL 02
58 +
59 SQRT
60 SQRT
61 RCL 01
62 *
63 -
64 RCL 00
65 -.375
66 Y↑X
67 RCL 02
68 .625
69 Y↑X
70 *
71 .8125
72 *
73 RCL 02
74 RCL 00
75 +
76 -.75
77 Y↑X
78 RCL 01
79 *
80 4
81 /
82 -
83 /
84 ST- 00
85 RCL 00
86 /
87 ABS
88 .25
89 X<=Y?
90 GTO 03
```

Output b.

```
91 RCL 00
92 .5
93 +
94 INT
95 "b"
96 XEQ "O"
97 GTO 02
```

Input subroutine

```
98♦LBL "IN"
99 CF 22
```

Program Listings

```
100 1
101 ST+ 00
102 RCL IND
00
103 "⊢="
104 ASTO Y
105 "⊢?"
106 CF 21
107 AVIEW
108 SF 21
109 CLA
110 ARCL Y
111 STOP
112 STO IND
00
113 FS? 22
114 FC? 55
115 RTN
116 ARCL X
117 PRA
118 RTN
119◆LBL "O"
120 "⊢="
121 ARCL X
122 AVIEW
```

Output subroutine.

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Data pointer/b	50	
	a/D		
	b/a		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 003 TOT. REG. 35 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EQUATIONS OF STATE

(Requires one memory module)

This program provides both ideal gas and Redlich-Kwong equations of state. Given four of the five state variables, the fifth is calculated. For the Redlich-Kwong solution, the critical pressure and temperature of the gas must be known. They are not needed for ideal gas solutions.

Values of the Universal Gas Constants

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/m^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.7302	atm-ft^3/lb mole-°R	atm	ft^3/lb mole	°R
10.73	psi-ft^3/lb mole-°R	psi	ft^3/lb mole	°R
1545	psf-ft^3/lb mole-°R	psf	ft^3/lb mole	°R

Critical Temperatures and Pressures

Substance	T_c, K	T_c, °R	P_c, ATM
Ammonia	405.6	730.1	112.5
Argon	151	272	48.0
Carbon dioxide	304.2	547.6	72.9
Carbon monoxide	133	239	34.5
Chlorine	417	751	76.1
Helium	5.3	9.5	2.26
Hydrogen	33.3	59.9	12.8
Nitrogen	126.2	227.2	33.5
Oxygen	154.8	278.6	50.1
Water	647.3	1165.1	218.2
Dichlorodifluoromethane	384.7	692.5	39.6
Dichlorofluoromethane	451.7	813.1	51.0
Ethane	305.5	549.9	48.2
Ethanol	516.3	929.3	63
Methanol	513.2	923.8	78.5
n-Butane	425.2	765.4	37.5
n-Haxane	507.9	914.2	29.9
n-Pentane	469.5	845.1	33.3
n-Octane	568.6	1023.5	24.6
Trichlorofluoromethane	471.2	848.1	43.2

Equations:

Ideal gas:

$$PV = nRT$$

Redlich-Kwong:

$$P = \frac{nRT}{(V - b)} - \frac{a}{T^{1/2}\, V\, (V + b)}$$

$$a = 4.934\ b\ nRT_c^{\ 1.5}$$

$$b = 0.0867\ \frac{nRT_c}{P_c}$$

where:

P is the absolute pressure;

V is the volume;

n is the number of moles present;

R is the universal gas constant;

T is the absolute temperature;

T_c is the critical temperature;

P_c is the critical pressure.

Remarks:

P, V, n and T must have units compatible with R.

At low temperatures or high pressures, the ideal gas law does not represent the behavior of real gases.

No equation of state is valid for all substances or over an infinite range of conditions. The Redlich-Kwong equation gives moderate to good accuracy for a variety of substances over a wide range of conditions. Results should be used with caution and tempered by experience.

Solutions for V, n, R and T, using the Redlich-Kwong equation, require an iterative technique. Newton's method is employed using the ideal gas law to generate the initial guess. Iteration time is generally a function of the amount of deviation from ideal gas behavior. For extreme cases, the routine may fail to converge entirely, resulting in "DATA ERROR".

Example 1:

0.63 g moles of air are enclosed in a 25,000 cm^3 space at 1200 K. What is the pressure in bars? Assume ideal gas behavior.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[XEQ] [ALPHA] ID [ALPHA]	P?
0 [R/S]	V?
25000 [R/S]	N?
0.63 [R/S]	R?
83.14 [R/S]	T?
1200 [R/S]	P=2.51

Example 2:

The specific volume of a gas in a container is 800 cm^3/g mole. The temperature will reach 400 K. What will the pressure be, according to the Redlich-Kwong relation?

$$P_c = 48.2 \text{ atm}$$

$$T_c = 305.5 \text{ K}$$

$$R = 82.05 \text{ cm}^3 - \text{atm/g mole-K}$$

Keystrokes:	Display:
[XEQ] [ALPHA] RK [ALPHA]	TC?
305.5 [R/S]	PC?
48.2 [R/S]	P?
0 [R/S]	V?
800 [R/S]	N?
1 [R/S]	R?
82.05 [R/S]	T?
400 [R/S]	P=36.29

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Select equation:			
	Ideal gas		[XEQ] ID	P?
	Redlich-Kwong		[XEQ] RK	TC?
3	For Redlich-Kwong only	T_c	[R/S]	PC?
		P_c	[R/S]	P?
4	Input appropriate values	P	[R/S]	V?
	(Zero for the unknown)	V	[R/S]	N?
		n	[R/S]	R?
		R	[R/S]	T?
		T	[R/S]	(unknown) =

Program Listings

Initialization

```
01♦LBL "ID"
02 0
03 SF 00
04 GTO 00
05♦LBL "RK"
06 1
07 CF 00
08 "TC?"
09 PROMPT
10 STO 13
11 "PC?"
12 PROMPT
13 STO 14
14♦LBL 00
15 SF 02
16 CF 01
17 FIX 2
18 "P?"
19 PROMPT
20 5
21 XEQ 00
22 "V?"
23 PROMPT
24 6
25 XEQ 00
26 "N?"
27 PROMPT
28 7
29 XEQ 00
30 "R?"
31 PROMPT
32 8
33 XEQ 00
34 "T?"
35 PROMPT
36 CF 02
37 9
38♦LBL 00
39 CF 01
40 STO 01
41 RDN
42 STO IND
01
43 X≠0?
44 GTO 01
45 R↑
46 STO 10
47 1
48 STO IND
01
49♦LBL 01
50 FS? 02
51 RTN
52 GTO IND
10
```

Calculate unknown

```
53♦LBL 05
54 "P="
55 GTO 00
56♦LBL 06
57 "V="
58♦LBL 00
59 RCL 07
60 RCL 08
61 *
62 RCL 09
63 *
64 RCL 05
65 RCL 06
66 *
67 /
68 STO IND
10
69 GTO 00
70♦LBL 07
71 SF 01
72 "N="
73 GTO 01
74♦LBL 08
75 SF 01
76 "R="
77 GTO 01
78♦LBL 09
79 "T="
80 SF 01
81♦LBL 01
82 RCL 05
83 RCL 06
84 *
85 RCL 07
86 /
87 RCL 08
88 /
89 RCL 09
90 /
91 STO IND
10
92♦LBL 00
93 FS? 00
94 GTO 10
95 XEQ 01
96 GTO 00
97♦LBL 02
```

Program Listings

```
 98 FS? 01
 99 XEQ 01
```

If ideal, display

```
100♦LBL 00
101 RCL 00
102 RCL 09
103 *
```

Calculate using Redlich-Kwong equations

```
104 RCL 06
105 RCL 12
106 -
107 STO 04
108 /
109 RCL 11
110 RCL 09
111 SQRT
112 /
113 STO 02
114 RCL 06
115 /
116 LASTX
117 RCL 12
118 +
119 STO 03
120 /
121 -
122 RCL 05
123 -
124 XEQ IND
10
125 /
126 ST- IND
10
127 RCL IND
10
128 /
129 ABS
130 1 E-4
131 X<=Y?
132 GTO 02
133 RCL IND
10
134 GTO 10
135♦LBL 06
```

$\frac{\partial P}{\partial V}$

```
136 RCL 06
137 ENTER↑
138 +
139 RCL 12
140 +
141 RCL 02
142 *
143 RCL 03
144 RCL 06
145 *
146 X↑2
147 /
148 RCL 00
149 RCL 09
150 *
151 RCL 04
152 X↑2
153 /
154 -
155 RTN
```

$\frac{\partial P}{\partial T}$

```
156♦LBL 09
157 RCL 00
158 RCL 04
159 /
160 RCL 02
161 2
162 /
163 RCL 09
164 /
165 RCL 06
166 /
167 RCL 03
168 /
169 +
170 RTN
```

$\frac{\partial P}{\partial n}$ or $\frac{\partial P}{\partial R}$

```
171♦LBL 07
172♦LBL 08
173 RCL 09
174 RCL 06
175 *
176 RCL 04
177 X↑2
178 /
179 RCL 06
180 ENTER↑
181 +
182 RCL 12
183 +
184 RCL 00
185 /
186 RCL 06
187 /
188 RCL 03
189 X↑2
190 /
191 RCL 02
192 *
193 -
194 RCL 00
195 *
```

Program Listings

```
196 RCL IND
10
197 /
198 RTN
199+LBL 05
200 LASTX
201 +
202 STO 05
203 GTO 10
```

Calculate a, b

```
204+LBL 01
205 RCL 07
206 RCL 08
207 *
208 STO 00
209 .0867
210 RCL 14
211 /
212 X<>Y
213 RCL 13
214 *
215 *
216 STO 12
217 LASTX
218 *
219 RCL 13
220 SQRT
221 *
222 4.934
223 *
224 STO 11
225 RTN
```

Display

```
226+LBL 10
227 ARCL X
228 AVIEW
229 STOP
230 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	NR	50	
	temp storage index		
	$a/T^{1/2}$		
	(V + b)		
	(V - b)		
05	P	55	
	V		
	n		
	f		
	T		
10	control	60	
	a		
	b		
	T_c		
	P_c		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 61 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Ideal	Redlich-Kwong
01		Calc a, b	
02		Inputting data	Calculate

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BLACK BODY THERMAL RADIATION

(Requires one memory module)

Bodies with finite temperatures emit thermal radiation. The higher the absolute temperature, the more thermal radiation emitted. Bodies which emit the maximum possible amount of energy at every wavelength for a specified temperature are said to be black bodies. While black bodies do not actually exist in nature, many surfaces may be assumed to be black for engineering considerations.



Figure 1.

Notes:

A half minute or more may be required to obtain $E_{b(0-\lambda)}$ or $E_{b(\lambda_1-\lambda_2)}$ since the integration is numerical.

Sources differ on values for constants. This could yield small discrepancies between published tables and program outputs.

Figure 1 is a representation of black body thermal emission as a function of wavelength. Note that as temperature increases, the area under the curves (total emissive power $E_{b(0-\infty)}$) increases. Also note that the wavelength of maximum emissive power λ_{max} shifts to the left as temperature increases.

This program calculates the wavelength of maximum emissive power for a given temperature, the temperature for which a given wavelength would be the wavelength of maximum emissive power, the total emissive power over all wavelengths, the emissive power at a particular wavelength, the emissive power from zero to a specified wavelength, and the emissive power between specified wavelengths.

Equations:

$$\lambda_{max} T_{\lambda_{max}} = c_3$$

$$E_{b(0-\infty)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)}$$

$$E_{b(0-\lambda)} = \int_0^\lambda E_{b\lambda} \, d\lambda$$

$$= 2\pi c_1 \sum_{k=i}^{\infty} -T/kc_2 \, e^{-\frac{kc_2}{T\lambda}} \left[\left(\frac{1}{\lambda}\right)^3 + \frac{3T}{\lambda^2 kc_2} + \frac{6}{\lambda}\left(\frac{T}{kc_2}\right)^2 + 6\left(\frac{T}{kc_2}\right)^3 \right]$$

$$E_{b(\lambda_1-\lambda_2)} = E_{b(0-\lambda_2)} - E_{b(0-\lambda_1)}$$

where:

λ_{max} is the wavelength of maximum emissivity in microns;

T is the absolute temperature in °R or K;

$E_{b(0-\infty)}$ is the total emissive power in Btu/hr-ft^2 or Watts/cm^2;

$E_{b\lambda}$ is the emissive power at λ in Btu/hr-ft^2-μm or Watts/cm^2-μm;

$E_{b(0-\lambda)}$ is the emissive power for wavelengths less than λ in Btu/hr-ft^2 or Watts/cm^2;

$E_{b(\lambda_1-\lambda_2)}$ is the emissive power for wavelengths between λ_1 and λ_2 in Btu/hr-ft^2 or Watts/cm^2.

$c_1 = 1.8887982 \times 10^7$ Btu-μm^4/hr-ft^2
$= 5.9544 \times 10^3$ Wμm^4/cm^2

$c_2 = 2.58984 \times 10^4$ μm-°R $= 1.4388 \times 10^4$ μm-K

$c_3 = 5.216 \times 10^3$ μm-°R $= 2.8978 \times 10^3$ μm-K

$\sigma = 1.713 \times 10^{-9}$ Btu/hr-ft^2-°R^4 $= 5.6693 \times 10^{-12}$ W/cm^2-K^4

$\sigma_{exp} = 1.731 \times 10^{-9}$ Btu/hr-ft^2-°R^4 $= 5.729 \times 10^{-12}$ W/cm^2-K^4

References: HP-67/97 Users' Library Program.

Example:

What percentage of the radiant output of a lamp is in the visible range (0.4 to 0.7 microns) if the filament of the lamp is assumed to be a black body at 2400K?

Keystrokes: (SIZE ⩾ 009)	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] BB [ALPHA]	UNITS?
SI [R/S]	TEMP?
2400 [R/S]	WAVELENGTH?
.4 [R/S]	SOLVE
[F]	WV LNTH 2?
.7 [R/S]	EbL-L=4.9679
[C]	EbTOT=188.0938
[÷]	0.0264
100 [x]	2.6412

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode.		[USER]	
2.	Initialize program		[XEQ] BB	UNITS?
3.	Input code for desired units SI, or EN	SI	[R/S]	
	or EN	EN	[R/S]	TEMP?
4.	Input temperature, if temperature is	Temp	[R/S]	WAVELENGTH?
	unknown, press [R/S].			
5.	Input first wavelength, if wavelength is	λ	[R/S]	SOLVE
	unknown, press [R/S].			
6.	Calculate any or all of the following:			
	λmax for a given temperature		[A]	WLMAX=
	T such that λ is λmax for T		[B]	TEMP=
	total emissive power at T		[C]	EbTOT=
	emissive power at T and λ		[D]	EbL=
	emissive power between zero and λ		[E]	Eb0-L=
	emissive power between λ_1 and λ_2		[F]	WV LNTH 2?
		λ_2	[R/S]	EbL-L=
7.	For a new case, go to step 2.			

Program Listings

01♦LBL "BB"	
02 CLRG	Initialize and
03 CF 22	prompt for units
04 "UNITS?"	
05 AON	
06 PROMPT	
07 AOFF	
08 ASTO X	
09 GTO IND X	
10♦LBL "SI"	
11 5954.4	Store units
12 STO 01	
13 14388	
14 STO 02	
15 2897.8	
16 STO 03	
17 5.6693 E-12	
18 STO 04	
19 GTO 00	
20♦LBL "EN"	
21 18887982	
22 STO 01	
23 25998.4	
24 STO 02	
25 5216	
26 STO 03	
27 .171312 E-08	
28 STO 04	
29♦LBL 00	Input prompting
30 "TEMP?"	
31 PROMPT	
32 STO 05	
33 "WAVELENGTH?"	
34 PROMPT	
35 STO 06	
36 "SOLVE"	
37 PROMPT	Calculate λmax
38♦LBL A	
39 RCL 03	
40 RCL 05	
41 /	
42 "WL MAX="	
43 ARCL X	
44 PROMPT	
45♦LBL B	
46 RCL 03	

47 RCL 06	
48 /	Calculate T(λmax)
49 "TEMP="	
50 ARCL X	
51 PROMPT	
52♦LBL C	Calculate E_b
53 RCL 05	total
54 X↑2	
55 X↑2	
56 RCL 04	
57 *	
58 "EbTOT="	
59 ARCL X	
60 PROMPT	
61♦LBL D	Calculate $E_{b\lambda}$
62 RCL 01	
63 ENTER↑	
64 +	
65 PI	
66 *	
67 RCL 06	
68 5	
69 Y↑X	
70 /	
71 RCL 02	
72 RCL 06	
73 /	
74 RCL 05	
75 /	
76 E↑X	
77 1	
78 -	
79 /	
80 "EbL="	
81 ARCL X	
82 PROMPT	
83♦LBL E	Calculate $E_b(0-\lambda)$
84 0	
85 STO 08	
86 STO 07	
87♦LBL 01	
88 RDN	
89 CLX	
90 RCL 08	
91 RCL 02	
92 RCL 05	
93 /	
94 -	
95 STO 08	
96 3	
97 X<>Y	

Program Listings

```
 98 /
 99 RCL 06
100 X↑2
101 /
102 LASTX
103 1/X
104 RCL 06
105 /
106 -
107 6
108 RCL 06
109 /
110 RCL 08
111 X↑2
112 /
113 -
114 6
115 RCL 08
116 X↑2
117 /
118 RCL 08
119 /
120 +
121 RCL 08
122 RCL 06
123 /
124 E↑X
125 *
126 RCL 08
127 /
128 ST+ 07
129 RCL 07
130 /
131 1 E-05
132 X<=Y?
133 GTO 01
134 RDN
135 CLX
136 RCL 07
137 ENTER↑
138 +
139 PI
140 *
141 RCL 01
142 *
143 FS?C 00
144 RTN
145 "Eb0-L="
146 ARCL X
147 PROMPT
148♦LBL F
```

Calculate $E_b(\lambda_1-\lambda_2)$

```
149 "WV LNTH
 2?"
150 PROMPT
151 ENTER↑
152 ENTER↑
153 SF 00
154 XEQ E
155 X<>Y
156 RCL 06
157 STO 00
158 RDN
159 STO 06
160 SF 00
161 XEQ E
162 -
163 ABS
164 RCL 00
165 STO 06
166 RDN
167 "EbL-L="
168 ARCL X
169 PROMPT
170 .END.
```

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	λ	50	
	C_1		
	C_2		
	C_3		
	σ		
05	T	55	
	λ,λ'		
	sum		
	kc_2/T		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 57 USER MODE
ENG ____ FIX 4 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Used	
22		Used	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

PSYCHROMETRIC PROPERTIES

(Requires two memory modules)

Given pressure, drybulb temperature and either wet bulb temperature or relative humidity or vapor pressure, this program computes wet bulb temperature, dew point temperature, vapor pressure, humidity ratio, relative humidity, enthalpy and specific volume.

Either English or SI units may be used in problem solution.

Variable	English Units	SI Units
Temperature	°F	°C
Pressure	psi	kPa
Humidity Ratio	lb_m/lb_m dry air	kg/kg dry air
Relative Humidity	%	%
Enthalpy	Btu/lb_m dry air	kJ/kg dry air
Specific Volume	ft^3/lb_m dry air	dm^3/kg dry air

Temperatures may range from -40°F (-40°C) to 300 °F (150°C). For temperatures less than the ice point, vapor is assumed in equilibrium with ice.

Total pressure may range from 1.0 psi (7.0 kPa) to 50.0 psi (345.0 kPa).

Relative humidity may range between 0.0 and 100%.

Specific humidity must be less than 0.2.

If limits are exceeded program will halt with alpha diagnostic in Display.

Equations:

$$W = \frac{c_{p,a}(t_{wb} - t_{db}) + \omega_{sat,wb}(h_{g,wb} - h_{f,wb})}{h_{v,db} - h_{f,wb}}$$

$$c_{p,a} = 0.240\ Btu/lb_m{}^\circ F = 1.0048\ kJ/kg^\circ C$$

$$h_v = h_g = a + bt \qquad a = 1061.0\ Btu/lb_m = 2501.0\ kJ/kg$$

$$b = 0.445\ Btu/lb_m{}^\circ F = 1.8631\ kJ/kg^\circ C$$

IF $t \geq 32^\circ F$ or $0^\circ C$ THEN $h_f = (t - c)d$

$$c = 32^\circ F = 0^\circ C$$

$$d = 1 Btu/lb_m{}^\circ F = 4.1868\ kJ/kg^\circ C$$

IF $t \geqslant 32°F$ or $0°C$ THEN $h_f = h_i = [(t-c)0.467 + e]d$

$e = -143.956°F = -79.97556°C$

$H = c_{p,a} t_{db} + W h_{v,db}$

$$W = \frac{R_{air}}{R_{vapor}} = \frac{PV}{P - PV} = \frac{R_{air}}{R_{vapor}} = 0.622$$

(also used to calculate P_{vapor})

$$P_{sat} = f\, e^{g/(t+h)}$$

t = temperature, °F or °C

e = 2.718282

IF $t \geqslant 32°F$ or $0°C$ THEN $f = 2.04466 \cdot 10^6$ psi $= 1.40974 \cdot 10^7$ kPa

$g = -7071.3°F = -3928.5°C$

$h = 385°F = 231.667°C$

IF $t \geqslant 32°F$ or $0°C$ THEN $f = 5.24506 \cdot 10^8$ psi $= 3.61633 \cdot 10^9$ kPa

$g = -11071°F = -6150.6°C$

$h = 460°F = 273.33°C$

$$V = \frac{R_{air}\; T}{P_{total} - P_{vapor}}$$

REL = Pvapor/Psat,db

where:

P	is the total pressure.
PV	is the vapor pressure
t_{db}	is the day bulb temperature.
P_{sat}	is the saturation pressure.
t_{wb}	is the wet bulb temperature.
t_{dp}	is the dew point temperature.
W	is the humidity ratio.
REL	is the relative humidity.
h	is the enthalpy.
V	is the specific volume.

Reference:

HP-67 Users' Library Program 867D, Psychrometric Properties, Donald H. Madsen.

Example 1:

P = 101.325 kPa, T_{wb} = 20°C T_{db} = 25°

Keystrokes: (SIZE ⩾ 010)	Display:
[///] [FIX] 4	
[XEQ] [ALPHA] PSYCHSI [ALPHA]	P=?
101.325 [R/S]	Tdb=?
25 [R/S]	PSAT=3.1762
[R/S]*	SELECT KEY: TWb REL PV
[A]	TWb=?
20 [R/S]	TdP=17.6023
[R/S]*	PV=2.0166
[R/S]*	W=0.0126
[R/S]*	REL=63.4931
[R/S]*	H=57.2983
[R/S]*	V=860.7494
[R/S]*	P=?

*Press [R/S] if you are not using a printer.

Example 2:

P = 25 psi, Tdb = 70.0°F, PV = 0.182 psi

Keystrokes:	Display:
[///] [FIX] 4	
[XEQ] [ALPHA] PSYCHE [ALPHA]	P=?
25 [R/S]	Tdb=?
70 [R/S]	PSAT=0.3640
[R/S]*	SELECT KEY: TWb REL PV
[E]	PV=?
.182 [R/S]	TWb=60.8398
[R/S]*	TdP=50.5726
[R/S]*	PV=0.1820
[R/S]*	W=0.0046
[R/S]*	REL=49.9988
[R/S]*	H=21.7817
[R/S]*	V=7.8987
[R/S]*	P=?

*Press [R/S] if you are not using a printer.

User Instructions

				SIZE: 010
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Initialize program for English units, or		[XEQ] PSYCHE	
	SI units.		[XEQ] PSYCHSI	P=?
3.	Key in total pressure.	P	[R/S]	Tdb=?
4.	Key in dry bulb temperature.	Tdb	[R/S]	PSAT=
			[R/S]*	SELECT KEY:
				TWb REL PV
5.	Press [A] if you know wet bulb temperature,		[A]	TWb=?
	[C] if you know relative humidity, or [E]		[B]	REL=?
	if you know vapor pressure, then go to		[C]	PV=?
	6a, 6b, or 6c.			
6a.	Key in wet bulb temperature and start			
	calculation.	TWb	[R/S]	
6b.	Key in relative humidity and start			
	calculation.	REL	[R/S]	TWb=
6c.	Key in vapor pressure and start calculation	PV	[R/S]	TWb=
7.	See outputs of:			
	Dew Point temperature,		[R/S]*	TdP=
	Vapor Pressure,		[R/S]*	PV=
	Humidity Ratio,		[R/S]*	W=
	Relative Humidity,		[R/S]*	REL=
	Enthalpy,		[R/S]*	H=
	Specific Volume		[R/S]*	V=
			[R/S]*	P=?
8.	For a new case using the same units, go to			
	step 3. Key in only values which change.			
*	Press [R/S] if you are not using a printer			

Program Listings

Initialize for SI units.

```
 01♦LBL "PSY
CHSI"
 02 SF 00
 03 GTO 00
```

Initialize for English units.

```
 04♦LBL "PSY
CHE"
 05 CF 00
```

Input T and T_{db}.

```
 06♦LBL 00
 07 0
 08 STO 00
 09 ADV
 10 "P"
 11 XEQ "IN"
 12 "Tdb"
 13 XEQ "IN"
 14 XEQ F
```

Calculate P_{sat}.

```
 15 ADV
 16 "PSAT"
 17 XEQ "O"
 18 STO 06
```

Prompt user to select key [A], [C], or [E].

```
 19 ADV
 20 "SELECT
KEY:"
 21 CF 21
 22 AVIEW
 23 SF 21
 24 PSE
 25♦LBL 22
 26 SF 27
 27 "TWb  RE
L  PV"
 28 PROMPT
 29 GTO 22
```

Input T_{wb}.

```
 30♦LBL A
 31 2
 32 STO 00
 33 "TWb"
 34 XEQ "IN"
 35 XEQ F
 36 XEQ 03
 37 "TWb HIG
H"
 38 X<0?
 39 PROMPT
 40 XEQ 04
 41 "TWb LOW
"
 42 X<0?
 43 PROMPT
 44 STO 07
 45 .2
 46 "SPC HUM
>=.2"
 47 X<=Y?
 48 PROMPT
 49 RDN
 50 RCL 01
 51 *
 52 RCL 07
 53 .622
 54 +
 55 /
 56 "TWb>Tdb
"
 57 XEQ 05
 58 ADV
 59 GTO 18
```

Input relative humidity.

```
 60♦LBL C
 61 3
 62 STO 00
 63 "REL"
 64 XEQ "IN"
 65 ADV
 66 RCL 02
 67 STO 03
 68 XEQ F
 69 X<>Y
 70 %
 71 STO 08
 72 GTO 00
```

Input vapor pressure.

```
 73♦LBL E
 74 RCL 02
 75 STO 03
 76 7
 77 STO 00
 78 "PV"
 79 XEQ "IN"
 80 "VP>Tdb"
 81 ADV
 82 XEQ 05
 83 RCL 08
 84♦LBL 00
 85 XEQ 03
 86 "VP>P"
 87 X<0?
 88 PROMPT
 89 STO 07
 90 .2
 91 "W>.2"
 92 X<=Y?
 93 PROMPT
 94 FS? 00
```

Program Listings

```
 95 80
 96 FC? 00
 97 140
 98 STO 09
 99 CF 02
```

Calculate T_{wb}.

```
100♦LBL 01
101 RCL 09
102 FS?C 02
103 CHS
104 ST- 03
105 2
106 ST/ 09
107 RCL 03
108 XEQ F
109 XEQ 03
110 X<0?
111 GTO 01
112 XEQ 04
113 RCL 07
114 -
115 X<0?
116 SF 02
117 ABS
118 1 E-5
119 X<=Y?
120 GTO 01
121 GTO 08
```

Calculate enthalpy difference.

```
122♦LBL 06
123 FC? 00
124 .445
125 FS? 00
126 1.8631
127 *
128 FC? 00
129 1061
130 FS? 00
131 2501
132 +
133 X<>Y
134 32
135 FS? 00
136 CLX
137 -
138 X<0?
139 XEQ 07
140 FC? 00
141 1
142 FS? 00
143 4.1868
144 *
145 -
146 RTN
147♦LBL 07
148 .467
149 *
150 FC? 00
151 -143.956
152 FS? 00
153 -79.9755
6
154 +
155 RTN
```

Relative humidity calculation.

```
156♦LBL 05
157 STO 08
158 RCL 06
159 /
160 1
161 X<>Y
162 X>Y?
163 PROMPT
164 1 E2
165 *
166 STO 04
167 RTN
```

Print results.

```
168♦LBL 08
169 RCL 03
170 "TWb"
171 XEQ "O"
172♦LBL 18
173 RCL 08
174 X=0?
175 GTO 09
176 32
177 FS? 00
178 CLX
179 XEQ F
180 X>Y?
181 GTO 00
182 RCL 08
183 FS? 00
184 1.40974
E7
185 FC? 00
186 2.04466
E6
187 /
188 LN
189 1/X
190 FS? 00
191 -3928.5
192 FC? 00
193 -7071.3
```

Program Listings

```
194 *
195 FS? 00
196 231.667
197 FC? 00
198 385
199 -
200 GTO 11
201♦LBL 00
202 RCL 08
203 FS? 00
204 3.61633
E9
205 FC? 00
206 5.24506
E8
207 /
208 LN
209 1/X
210 FS? 00
211 -6150.6
212 FC? 00
213 -11071
214 *
215 FS? 00
216 273.33
217 FC? 00
218 460
219 -
220 GTO 11
```

Saturation pressure subroutine.

```
221♦LBL F
222 32
223 FS? 00
224 CLX
225 X>Y?
226 GTO 00
227 RDN
228 FS? 00
229 231.667
230 FC? 00
231 385
232 +
233 FS? 00
234 -3928.5
235 FC? 00
236 -7071.3
237 X<>Y
238 /
239 E↑X
240 FS? 00
241 1.40974
E7
242 FC? 00
243 2.04466
E6
244 *
245 RTN
```

Saturation pressure for sub-icepoint conditions.

```
246♦LBL 00
247 RDN
248 FS? 00
249 273.33
250 FC? 00
251 460
252 +
253 FS? 00
254 -6150.6
255 FC? 00
256 -11071
257 X<>Y
258 /
259 E↑X
260 FS? 00
261 3.61633
E9
262 FC? 00
263 5.24506
E8
264 *
265 RTN
```

Calculate specific humidity from vapor pressure.

```
266♦LBL 03
267 "WSAT OR
 VP=P"
268 .622
269 X<>Y
270 *
271 RCL 01
272 LASTX
273 -
274 X=0?
275 PROMPT
276 /
277 RTN
```

Calculate specific humidity from wet and dry bulb temperatures.

```
278♦LBL 04
279 RCL 03
280 RCL 03
281 XEQ 06
282 *
283 RCL 03
284 RCL 02
285 -
286 FC? 00
287 .24
288 FS? 00
289 1.0048
```

Program Listings

```
290 *
291 +
292 RCL 03
293 RCL 02
294 XEQ 06
295 /
296 RTN
```

Output values.

```
297♦LBL 11
298 "TdP"
299 XEQ "O"
300♦LBL 09
301 RCL 08
302 "PV"
303 XEQ "O"
304 RCL 07
305 "W"
306 XEQ "O"
307 RCL 04
308 "REL"
309 XEQ "O"
310 RCL 02
311 FC? 00
312 .24
313 FS? 00
314 1.0048
315 *
316 32
317 FS? 00
318 ST- X
319 RCL 02
320 XEQ 06
321 RCL 07
322 *
323 +
324 "H"
325 XEQ "O"
326 RCL 02
327 FC? 00
328 459.67
329 FS? 00
330 273.15
331 +
332 FC? 00
333 .3701
334 FS? 00
335 286.7
336 *
337 RCL 01
338 RCL 08
339 -
340 /
341 "V"
342 XEQ "O"
343 GTO 00
```

Input subroutine.

```
344♦LBL "IN"
345 CF 22
346 1
347 ST+ 00
348 RCL IND
00
349 "⊢="
350 ASTO Y
351 "⊢?"
352 CF 21
353 AVIEW
354 SF 21
355 CLA
356 ARCL Y
357 STOP
358 STO IND
00
359 FS? 22
360 FC? 55
361 RTN
362 ARCL X
363 PRA
364 RTN
```

Output subroutine

```
365♦LBL "O"
366 "⊢="
367 ARCL X
368 AVIEW
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Data Pointer	50	
	P		
	Tdb		
	TWb		
	Used		
05	VP	55	
	PSAT		
	W		
	PV		
	Used		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 138 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		SI	English
21		Stop on view	
27		User mode	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEAT EXCHANGERS

(Requires one memory module)

This program allows analysis of counterflow, parallel flow, parallel-counterflow, and crossflow heat exchangers.

Figure 1:



Counter-Flow

Figure 2:



Parallel-Flow

Figure 3:



Parallel-Counter-Flow
(Even Number Of Tube Passes)

Figure 4:



Equations:

Heat exchanger effectiveness E is the ratio of actual heat transfer to maximum possible heat transfer.

$$E = \frac{Q}{C_{min}\ (T_{hin} - T_{cin})} = \frac{C_h (T_{hin} - T_{ho})}{C_{min}\ (T_{hin} - T_{cin})} = \frac{C_c (T_{co} - T_{cin})}{C_{min} (T_{hin} - T_{cin})}$$

where:

Q is the actual heat transfer.

T_{hin} and T_{cin} are the inlet temperatures of the hot and cold fluids respectively.

T_{ho} and T_{co} are the outlet temperatures of the hot and cold fluids respectively.

C_h and C_c are the heat capacities of the hot and cold fluids, respectively, e.g., $C_h = m_h \times c_{ph}$, where m_h is the flow rate and c_{ph} is the specific heat capacity of the hot fluid.

C_{min} and C_{max} (which are used later) are the smaller and larger values of C_h and C_c.

Effectiveness can be related to the product of the surface area of the heat exchanger and the overall heat transfer coefficient for specific geometries. This product is designated AU. The geometrics considered in this pac have the following correlations:

Counterflow (see figure 1)

$$E = \frac{1 - e^{-\frac{AU}{C_{min}}\left(1 - \frac{C_{min}}{C_{max}}\right)}}{1 - (C_{min}/C_{max})e^{-\frac{AU}{C_{min}}\left(1 - \frac{C_{min}}{C_{max}}\right)}}$$

For $C_{min}/C_{max} = 1$

$$E = \frac{AU/C_{min}}{1 + AU/C_{min}}$$

Parallel Flow (see figure 2)

$$E = \frac{1 - e^{-\frac{AU}{C_{min}}(1 + C_{min}/C_{max})}}{1 + C_{min}/C_{max}}$$

For $C_{min}/C_{max} = 0$, C_{min} is set to 1.

Parallel-Counterflow (well mixed with an even number of tube passes; see Figure 3)

$$E = \frac{2}{\left(1 + \frac{C_{min}}{C_{max}}\right) + \sqrt{1 + \left(\frac{C_{min}}{C_{max}}\right)^2}\left[\frac{1 + e^{-x}}{1 - e^{-x}}\right]}$$

where:

$$x = \frac{AU}{C_{min}} \sqrt{1 + \left(\frac{C_{min}}{C_{max}}\right)^2}$$

Crossflow (both fluids unmixed; see figure 4)

No exact expression exists for this case, but the following is a very good approximation. Note that an iterative solution is required for AU.

$$E = 1 - e^{\left(e^{\left(-\frac{AU}{C_{min}} \frac{C_{min}}{C_{max}} y\right)} - 1\right)\left(\frac{C_{max}}{C_{min}} \frac{1}{y}\right)}$$

where:

$$y = \left[\frac{C_{min}}{AU}\right]^{0.22}$$

References:

W.M. Kays and A.L. London, *Compact Heat Exchangers*, National Press, 1955

Eckert and Drake, *Heat and Mass Transfer*, McGraw-Hill.

Remarks:

For cases where the inlet and outlet temperatures of one of the fluids are equal(change of phase), use zero for the heat capacity of that fluid.

The solution for AU in the crossflow configuration takes significantly longer than other solutions because of the iterative technique required.

The program must be allowed to solve for all values (AU, Q, T_{co}, T_{ho}, and E). It is quite possible for the heat balance equations to yield physically meaningless solutions for a particular configuration. However, the message "2ND LAW ERR" will be displayed if the 2nd law of thermodynamics has been violated during the calculation of AU or Q.

This program is organized into five routines. The first routine performs heat balance calculations and acts as a controller for the four configuration subroutines. Each configuration subroutine has two sections that calculate AU and E for that heat exchanger. You should first load the controller, then load the configuration of interest as a separate program.

You may wish to write your own configuration routines. A routine for a configuration must be in the following format:

```
               LBL ACON
                  •
                  •
                  •
(calculates AU for this configuration)
                  •
                  •
                  •
                 RTN
                  •
                  •
                  •
               LBL ECON
                  •
                  •
                  •
(calculates E for this configuration)
                  •
                  •
                  •
                 END
```

Example:

A liquid at 168°F is to be cooled to 117°F. The liquid has a heat capacity of 0.42 Btu/LBM-°F and flows at 7700 LBM/hr. Cooling water (heat capacity = 1.00) is available at 4800 lbm/hr at 50°F. For counterflow, crossflow, parallel-counterflow, and parallel flow heat exchangers with overall coefficients of 55 Btu/hr-ft^2-°F what areas are required?

Keystrokes: (SIZE ⩾ 023)	Display:
[///] [FIX] 4	
Load main routine and counterflow subroutine.	
[XEQ] [ALPHA] HEATX [ALPHA]	TC IN=?
50 [R/S]	TH IN=?
168 [R/S]	MC=?
4800 [R/S]	MH=?
7700 [R/S]	CPC=?
1 [R/S]	CPH=?
.42 [R/S]	SELECT KEY: E AU Q TC TH
Since the temperature of the outgoing fluid is known, press the [E] key.	
[E]	THO=?
117 [R/S]	E=0.4322
[R/S]*	AU=2,198.7662
[R/S]*	Q=164,933.9999
[R/S]*	TCO=84.3612
[R/S]*	SELECT KEY: E AU Q TC TH

Keystrokes:	Display:
Since A = AU/U, calculate A.	
2198.7662 [ENTER] 55 [÷]	39.9776
Load crossflow subroutine.	
[XEQ] [ALPHA] HEATX [ALPHA]	TC IN=?
[R/S]	TH IN=?
[R/S]	MC=?
[R/S]	MH=?
[R/S]	CPC=?
[R/S]	CPH=?
[R/S]	SELECT KEY: E AU Q TC TH
[E]	THO=?
[R/S]	E=0.4322
[R/S]*	AU=2,353.6675
[R/S]*	Q=164,934.0000
[R/S]*	TCO=84.3613
[R/S]	SELECT KEY: E AU Q TC TH
2353.6675 [ENTER] 55 [÷]	42.7940

An analogus procedure will yield areas of 42.2776 ft^2 and 45.1494 ft^2 for parallel-counterflow and parallel exchanges respectively.

User Instructions

SIZE: 023

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and then:		[GTO] ..	
2.	Load configuration subroutine corresponding			
	to your heat exchanger geometry		[XEQ] HEATX	TC IN=?
3.	Input inlet temperature of cold fluid	T_{cin}	[R/S]	TH IN=?
4.	Input inlet temperature of hot fluid	T_{hin}	[R/S]	MC=?
5.	Input mass flow rate of cold fluid	m_c	[R/S]	MH=?
6.	Input mass flow rate of hot fluid	m_h	[R/S]	CPC=?
7.	Input specific heat of cold fluid	C_{pc}	[R/S]	CPH=?
8.	Input specific heat of hot fluid	C_{ph}	[R/S]	SELECT KEY
				E AU Q TC TH
9.	Select the known value:			
	heat exchanger effectiveness		[A]	E=?
	area-heat transfer coefficient product		[B]	AU=?
	heat transfer		[C]	Q=?
	outlet temperature of cold fluid		[D]	TCO=?
	outlet temperature of hot fluid		[E]	THO=?
	input the known value.	E	[R/S]	
		AU	[R/S]	
		Q	[R/S]	
		TCO	[R/S]	
		THO	[R/S]	
	The four variables other than the one you			E=
	input will be output. The output order		[R/S]*	AU=
	will vary depending on which value was		[R/S]*	Q=
	input. If the 2nd law of thermodynamics		[R/S]*	TCO=
	is violated, the message "2ND LAW ERR"		[R/S]*	THO=
	will be displayed.		[R/S]*	SELECT KEY

User Instructions

SIZE: 023

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
10.	For a new problem, go to step 2 or step 9.			E AU Q TC TH
	It is not necessary to key in any values			
	which do not change. Ignore the prompts			
	and press [R/S].			
*	Press [R/S] if you do not have a printer.			

Program Listings

Heat Exchanger - Main Routine

Input values.

```
 01♦LBL "HEA
TX"
 02 2
 03 STO 00
 04 "TC IN"
 05 XEQ "IN"
 06 "TH IN"
 07 XEQ "IN"
 08 14
 09 STO 00
 10 "MC"
 11 XEQ "IN"
 12 "MH"
 13 XEQ "IN"
 14 "CPC"
 15 XEQ "IN"
 16 RCL 15
 17 *
 18 STO 05
 19 "CPH"
 20 XEQ "IN"
 21 RCL 16
 22 *
 23 STO 06
 24 "CON"
 25 ASTO 22
```

Select last input.

```
 26♦LBL 06
 27 CF 02
 28 CF 03
 29 CF 04
 30 CF 05
 31 CF 06
 32 CF 21
 33 SF 27
 34 "SELECT
KEY:"
 35 AVIEW
 36 SF 21
 37 PSE
 38♦LBL 00
 39 ADV
 40 "E AU Q
TC TH"
 41 PROMPT
 42 GTO 00
```

Input E.

```
 43♦LBL A
 44 SF 02
 45 9
 46 STO 00
 47 SF 01
 48 "E"
 49 XEQ "IN"
 50 ADV
 51 GTO 16
```

Input AU.

```
 52♦LBL B
 53 SF 03
 54 10
 55 STO 00
 56 "AU"
 57 XEQ "IN"
 58 ADV
 59 GTO 01
```

Input Q.

```
 60♦LBL C
 61 SF 04
 62 11
 63 STO 00
 64 "Q"
 65 XEQ "IN"
 66 ADV
 67 GTO 05
```

Input TCO.

```
 68♦LBL D
 69 SF 05
 70 12
 71 STO 00
 72 "TCO"
 73 XEQ "IN"
 74 ADV
 75 GTO 14
```

Input THO.

```
 76♦LBL E
 77 SF 06
 78 13
 79 STO 00
 80 "THO"
 81 XEQ "IN"
 82 ADV
 83 GTO 04
```

Calculate AU.

```
 84♦LBL 16
 85 FS?C 03
 86 GTO 06
 87 RCL 10
 88 "A"
 89 XEQ 08
 90 STO 11
 91 "AU"
 92 XEQ "O"
```

Calculate Q.

```
 93♦LBL 01
 94 FS?C 04
 95 GTO 06
 96 RCL 11
 97 "E"
 98 XEQ 08
 99 RCL 07
```

Program Listings

Heat Exchanger - Main Routine

```
100 *
101 RCL 04
102 RCL 03
103 -
104 *
105 STO 12
106 "Q"
107 XEQ "O"
```

Calculate TCO.

```
108♦LBL 05
109 FS?C 05
110 GTO 06
111 RCL 12
112 RCL 05
113 X≠0?
114 /
115 RCL 03
116 +
117 STO 13
118 "TCO"
119 XEQ "O"
```

Calculate THO.

```
120♦LBL 14
121 FS?C 06
122 GTO 06
123 RCL 13
124 RCL 03
125 -
126 RCL 05
127 *
128 RCL 06
129 X≠0?
130 /
131 RCL 04
132 -
133 CHS
134 STO 14
135 "THO"
136 XEQ "O"
```

Calculate E.

```
137♦LBL 04
138 FS?C 02
139 GTO 06
140 RCL 13
141 RCL 03
142 -
143 RCL 05
144 *
145 RCL 04
146 RCL 14
147 -
148 RCL 06
149 *
150 X=0?
151 X<>Y
152 RCL 04
153 RCL 03
154 -
155 /
156 RCL 05
157 RCL 06
158 X<=Y?
159 X<>Y
160 RDN
161 X=0?
162 X<> T
163 /
164 STO 10
165 SF 01
166 "E"
167 XEQ "O"
168 GTO 16
```

A0 for C_{min}=0.00.

```
169♦LBL "A0"
170 1
171 RCL 10
172 -
173 LN
174 CHS
175 RTN
```

E0 for C_{min}=0.00.

```
176♦LBL "E0"
177 1
178 RCL 11
179 CHS
180 E↑X
181 -
182 RTN
```

Find C_{min} and execute configuration subroutine.

```
183♦LBL 08
184 RCL 05
185 RCL 06
186 X>Y?
187 X<>Y
188 X<>Y
189 STO 07
190 X<>Y
191 X≠0?
192 STO 07
193 X<>Y
194 X≠0?
195 /
196 STO 09
197 SF 25
198 X=0?
199 "⊢0"
200 X≠0?
201 ARCL 22
```

Program Listings

```
202 ASTO T
203 XEQ IND
T
204 FS?C 25
205 RTN
```

Trap errors from subroutines

```
206 "2ND LAW
 ERR"
207 PROMPT
208 GTO 06
```

Input subroutine

```
209♦LBL "IN"
210 CF 22
211 1
212 ST+ 00
213 RCL IND
00
214 "⊢="
215 ASTO Y
216 "⊢?"
217 CF 21
218 AVIEW
219 SF 21
220 CLA
221 ARCL Y
222 STOP
223 STO IND
00
224 FS? 22
225 FC? 55
226 RTN
```

Print if printer is attached

```
227 ARCL X
228 PRA
229 RTN
```

Output subroutine

```
230♦LBL "O"
231 "⊢="
232 ARCL X
233 AVIEW
234 .END.
```

Program Listings

Parallel Flow Subroutine

Calculate AU.

```
01♦LBL "ACO
N"
02 RCL 09
03 1
04 +
05 RCL 10
06 *
07 CHS
08 1
09 +
10 LN
11 CHS
12 1
13 RCL 09
14 +
15 /
16 RCL 07
17 *
18 RTN
```

Calculate E.

```
19♦LBL "ECO
N"
20 1
21 +
22 RCL 11
23 RCL 07
24 /
25 *
26 CHS
27 E↑X
28 CHS
29 1
30 +
31 1
32 RCL 09
33 +
34 /
35 RTN
```

Program Listings

Counter Flow Subroutine

Calculate AU.

```
01♦LBL "ACO
N"
02 RCL 10
03 1/X
04 -
05 1
06 LASTX
07 -
08 /
09 LN
10 1
11 RCL 09
12 -
13 X=0?
14 GTO 10
15 /
16 RCL 07
17 *
18 RTN
19♦LBL 10
20 RCL 10
21 1
22 RCL 10
23 -
24 /
25 RCL 07
26 *
27 RTN
```

Calculate E.

```
28♦LBL "ECO
N"
29 1
30 -
31 RCL 11
32 RCL 07
33 /
34 *
35 E↑X
36 1
37 X<>Y
38 -
39 LASTX
40 RCL 09
41 *
42 1
43 X<>Y
44 -
45 X=0?
46 GTO 11
47 /
48 RTN
49♦LBL 11
50 RCL 11
51 RCL 07
52 /
53 ENTER↑
54 ENTER↑
55 1
56 +
57 /
58 RTN
```

Program Listings

Parallel-Counter Flow Subroutine

Calculate AU.

```
01♦LBL "ACO
N"
02 XEQ 12
03 2
04 *
05 RCL 12
06 2
07 RCL 10
08 /
09 +
10 RCL 08
11 -
12 /
13 CHS
14 1
15 +
16 LN
17 RCL 12
18 /
19 CHS
20 RCL 07
21 /
22 LASTX
23 X↑2
24 *
25 RTN
```

Calculate E.

```
26♦LBL "ECO
N"
27 XEQ 12
28 RCL 11
29 RCL 07
30 /
31 RCL 12
32 *
33 CHS
34 E↑X
35 1
36 X<>Y
37 +
38 1
39 LASTX
40 -
41 /
42 RCL 12
43 *
44 RCL 08
45 +
46 2
47 X<>Y
48 /
49 RTN
```

```
50♦LBL 12
51 RCL 09
52 1
53 +
54 STO 08
55 RCL 09
56 X↑2
57 1
58 +
59 SQRT
60 STO 12
61 RTN
```

Program Listings

Cross Flow Subroutine

Calculate AU.

```
01♦LBL "ACON"
02 0
03 STO 19
04 1
05 RCL 10
06 CHS
07 STO 21
08 +
09 LN
10 CHS
11 STO 11
12♦LBL 13
13 RCL 11
14 XEQ "ECON"
15 RCL 10
16 -
17 STO 20
18 RCL 19
19 RCL 11
20 STO 19
21 -
22 RCL 21
23 RCL 20
24 STO 21
25 -
26 /
27 *
28 ST- 11
29 ABS
30 1 E-4
31 X<=Y?
32 GTO 13
33 RCL 11
34 RTN
```

Calculate E.

```
35♦LBL "ECON"
36 RCL 11
37 RCL 07
38 /
39 ENTER↑
40 ENTER↑
41 .22
42 Y↑X
43 RCL 09
44 /
45 /
46 LASTX
47 X<>Y
48 CHS
49 E↑X
50 1
51 -
52 *
53 E↑X
54 CHS
55 1
56 +
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Storage Index	50	
	TC IN		
	TH IN		
05	CC	55	
	CH		
	Cmin		
	Cmax		
	Cmin/Cmax		
10	E	60	
	AU		
	Q		
	Tco		
	Tho		
15	MC	65	
	MH		
	cpc		
	cph		
	(AU_{i-1})		
20	$F(AU_i)$	70	
	$F(AU_{i-1})$		
	"CON"		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 23 TOT. REG. 102 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
0		Unit conversion	
1	C		
2	C		
3	C		
4	C		
5	C		
6	C		
23		Alpha input	
55		Printer connected	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

DECIBEL ADDITION AND SUBTRACTION

This program adds or subtracts sound pressure levels measured in decibels.

Equations:

$$dB_1 + dB_2 = 10 \log (10^{dB_1/10} + 10^{dB_2/10})$$

$$dB_1 - dB_2 = 10 \log (10^{dB_1/10} - 10^{dB_2/10})$$

When subtracting, if $dB_1 < dB_2$, the program will exchange values.

Example 1:

A noise level of 72 decibels is measured in a room. The air conditioning is turned on and the noise level increases to 74 decibels. What is the noise level of the air conditioning system.

Keystrokes:	Display:
[///] [FIX] 2	
74 [ENTER] 72 [XEQ] [ALPHA] dB- [ALPHA]	69.67 dB

Example 2:

A compressor is known to have a sound pressure level of 90 dB. The background is 85 dB. What is the total?

Keystrokes:	Display:
[///] [FIX] 2	
90 [ENTER] 85 [XEQ] [ALPHA] dB+ [ALPHA]	91.19dB

User Instructions

SIZE: ANY

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Key in sound pressure levels in decibels:	dB_1	[ENTER ↑]	dB_1
		dB_2		
3a.	Add dB_1 and dB_2		[XEQ] dB+*	$(dB_1+dB_2)dB$
3b.	or subtract dB_2 from dB_1		[XEQ] dB-*	$(dB_1-dB_2)dB$
4.	For a new case go to step 2.			
	*Assigning dB+ to the shifted plus key and			
	dB- to the shifted minus enhances the			
	usability of this program.			

Program Listings

```
 01◆LBL "dB+
"
 02 XEQ 00
 03 +
 04 GTO 01
 05◆LBL "dB-
"
 06 XEQ 00
 07 -
 08 ABS
 09◆LBL 01
 10 LOG
 11 10
 12 *
 13 CLA
 14 ARCL X
 15 "⊦ dB"
 16 AVIEW
 17 RTN
 18◆LBL 00
 19 10
 20 ST/ Z
 21 /
 22 X<>Y
 23 10↑X
 24 X<>Y
 25 10↑X
```

01–04	Add decibels.
05–08	Subtract decibels.
09–17	Display result.
18–25	Convert for add or subtract.

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE ANY TOT. REG. 07 USER MODE
ENG ___ FIX ___ SCI ___ ON ___ OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
(suggested)			
dB+	-61		
dB-	-51		

TEMPERATURE CONVERSION

This program converts interchangeably between the four types of temperature.

Equations:

$K = (°F + 459.67)/1.8$

$K = °C + 273.15$

$K = °R / 1.8$

where:

K is temperature in Kelvins.

°F is temperature in degrees Fahrenheit.

°C is temperature in degrees Celsius.

°R is temperature in degrees Rankine.

Remarks:

Only the stack registers, LASTx and the alpha register are used in the conversions.

Example:

Convert: 472K to °R
27°F to °C
25°C to °F
100°C to °F

Keystrokes:	Display:	
[///] [FIX] 2		
[XEQ] [ALPHA] TCON [ALPHA]	F C R K	
472 [D]	F C R K	
[C]	849.60	(°R)
[R/S]	F C R K	
27 [A]	F C R K	
[B]	-2.78	(°C)
25 [B]	F C R K	
[A]	77	(°F)
100 [B]	F C R K	
[A]	212	(°F)

User Instructions

SIZE: ANY

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	Key in value to be converted.	T_{IN}		
3.	Initialize program		[XEQ] TCON	F C R K
4.	The display defines the top row keys for			
	input. Press [A] if the value in X is in			
	Fahrenheit, [B] for Celsius, [C] for			
	Rankine and [D] for Kelvin.		[A],[B],[C],or[D]	F C R K
5.	The display now defines the top row keys			
	for output. Select the unit of output			
	and press the corresponding key.		[A],[B],[C],or[D]	T_{OUT}
6.	For a new case, go to step 2. If you			
	remember which keys are which, skip step 3.			
	If you need a reminder, press [R/S] in			
	place of [XEQ] TCON. It is also per-			
	missible to interchange the order of			
	steps 2 and 3.			

Program Listings

01♦LBL 00 02 RTN	Display answer
03♦LBL "TCON" 04 SF 27 05 "F C R K" 06 PROMPT	Initialize
07♦LBL A 08 459.67 09 + 10 1.8 11 / 12 GTO D	°F-K
13♦LBL B 14 273.15 15 + 16 GTO D	°C-K
17♦LBL C 18 1.8 19 /	°R-K
20♦LBL D 21 PROMPT	K-K
22♦LBL A 23 1.8 24 * 25 459.67 26 - 27 GTO 00	K-°F
28♦LBL B 29 273.15 30 - 31 GTO 00	K-°C
32♦LBL C 33 1.8 34 *	K-°R
35♦LBL D 36 GTO 00 37 .END.	K-K

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE ANY TOT. REG. 14 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
27	S	User mode	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Heating, Ventilating & Air Conditioning

HEATING, VENTILATING & AIR CONDITIONING

OVERALL HEAT TRANSFER
COEFFICIENTS
PROGRAM REGISTERS NEEDED: 16

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 – 4)

ROW 2 (5 – 7)

ROW 3 (7 – 14)

ROW 4 (14 – 20)

ROW 5 (21 – 26)

ROW 6 (26 – 33)

ROW 7 (33 – 40)

ROW 8 (41 – 48)

ROW 9 (48 – 50)

INSULATION BREAK EVEN ANALYSIS

PROGRAM REGISTERS NEEDED: 29

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 - 5)

ROW 2 (5 - 8)

ROW 3 (8 - 11)

ROW 4 (11 - 13)

ROW 5 (14 - 19)

ROW 6 (20 - 24)

ROW 7 (25 - 36)

ROW 8 (37 - 46)

ROW 9 (47 - 57)

ROW 10 (58 - 63)

ROW 11 (64 - 74)

ROW 12 (74 - 79)

ROW 13 (79 - 85)

ROW 14 (86 - 93)

ROW 15 (93 - 98)

ROW 16 (99 - 99)

AIR FLOW IN CIRCULAR DUCTS

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

PROGRAM REGISTERS NEEDED: 90

ROW 1 (1 - 3)

ROW 2 (4 - 7)

ROW 3 (7 - 13)

ROW 4 (14 - 17)

ROW 5 (18 - 24)

ROW 6 (24 - 30)

ROW 7 (30 - 36)

ROW 8 (37 - 40)

ROW 9 (40 - 44)

ROW 10 (45 - 50)

ROW 11 (50 - 53)

ROW 12 (53 - 57)

ROW 13 (57 - 60)

ROW 14 (60 - 65)

ROW 15 (66 - 71)

ROW 16 (71 - 73)

ROW 17 (74 - 80)

ROW 18 (81 - 87)

ROW 19 (87 - 92)

ROW 20 (92 - 97)

ROW 21 (98 - 105)

ROW 22 (105 - 108)

ROW 23 (109 - 112)

ROW 24 (112 - 117)

ROW 25 (118 - 122)

ROW 26 (123 - 125)

ROW 27 (126 - 129)

ROW 28 (129 - 132)

ROW 29 (132 - 136)

ROW 30 (137 - 143)

ROW 31 (144 - 152)

ROW 32 (153 - 163)

ROW 33 (164 - 169)

ROW 34 (170 - 179)

ROW 35 (180 - 188)

ROW 36 (189 - 197)

ROW 37 (198 - 210)

ROW 38 (211 - 222)

ROW 39 (223 - 231)

ROW 40 (232 - 242)

ROW 41 (243 - 250)

ROW 42 (250 - 252)

ROW 43 (252 - 259)

ROW 44 (260 - 269)

ROW 45 (270 - 274)

ROW 46 (275 - 280)

ROW 47 (281 - 288)

ROW 48 (289 - 293)

ROW 49 (294 - 295)

AIR DUCT CONVERSION

PROGRAM REGISTERS NEEDED: 33

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 - 3)

ROW 2 (4 - 9)

ROW 3 (9 - 18)

ROW 4 (19 - 27)

ROW 5 (27 - 29)

ROW 6 (29 - 36)

ROW 7 (36 - 44)

ROW 8 (45 - 54)

ROW 9 (54 - 65)

ROW 10 (65 - 71)

ROW 11 (71 - 77)

ROW 12 (78 - 88)

ROW 13 (88 - 96)

ROW 14 (96 - 101)

ROW 15 (101 - 106)

ROW 16 (107 - 114)

ROW 17 (115 - 120)

ROW 18 (120 - 123)

PROGRAM REGISTERS NEEDED: 47

ROW 1 (1 - 5)

ROW 2 (5 - 10)

ROW 3 (11 - 17)

ROW 4 (18 - 24)

ROW 5 (25 - 30)

ROW 6 (30 - 37)

ROW 7 (38 - 47)

ROW 8 (48 - 55)

ROW 9 (55 - 65)

ROW 10 (66 - 73)

ROW 11 (73 - 79)

ROW 12 (80 - 91)

ROW 13 (91 - 98)

ROW 14 (98 - 108)

ROW 15 (109 - 121)

ROW 16 (122 - 130)

ROW 17 (130 - 138)

ROW 18 (139 - 151)

ROW 19 (152 - 164)

ROW 20 (165 - 177)

ROW 21 (178 - 190)

ROW 22 (191 - 202)

ROW 23 (203 - 210)

ROW 24 (211 - 222)

ROW 25 (222 - 230)

ROW 26 (230 - 230)

BLACK BODY THERMAL RADIATION

PROGRAM REGISTERS NEEDED: 50

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (10 - 13)

ROW 4 (13 - 17)

ROW 5 (17 - 20)

ROW 6 (20 - 23)

ROW 7 (23 - 27)

ROW 8 (27 - 30)

ROW 9 (30 - 33)

ROW 10 (33 - 36)

ROW 11 (36 - 42)

ROW 12 (42 - 49)

ROW 13 (49 - 52)

ROW 14 (53 - 58)

ROW 15 (58 - 61)

ROW 16 (62 - 74)

ROW 17 (75 - 82)

ROW 18 (83 - 94)

ROW 19 (95 - 107)

ROW 20 (108 - 120)

ROW 21 (121 - 131)

ROW 22 (131 - 140)

ROW 23 (141 - 146)

ROW 24 (146 - 149)

ROW 25 (149 - 157)

ROW 26 (158 - 167)

ROW 27 (167 - 170)

PSYCHROMETRIC PROPERTIES

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

PROGRAM REGISTERS NEEDED: 129

ROW 1 (1 : 2)

ROW 2 (3 : 5)

ROW 3 (5 : 12)

ROW 4 (12 : 16)

ROW 5 (16 : 20)

ROW 6 (20 : 25)

ROW 7 (25 : 27)

ROW 8 (27 : 33)

ROW 9 (33 : 37)

ROW 10 (37 : 40)

ROW 11 (41 : 45)

ROW 12 (46 : 47)

ROW 13 (48 : 56)

ROW 14 (56 : 59)

ROW 15 (60 : 65)

ROW 16 (66 : 74)

ROW 17 (75 : 80)

ROW 18 (80 : 85)

ROW 19 (86 : 91)

ROW 20 (91 : 97)

ROW 21 (98 : 106)

ROW 22 (107 : 112)

ROW 23 (113 : 120)

ROW 24 (121 : 126)

ROW 25 (126 : 130)

ROW 26 (131 : 138)

ROW 27 (139 : 143)

ROW 28 (143 : 151)

ROW 29 (151 : 153)

ROW 30 (153 : 161)

ROW 31 (162 : 170)

ROW 32 (170 : 177)

ROW 33 (177 : 184)

ROW 34 (184 : 186)

ROW 35 (186 : 191)

ROW 36 (191 : 193)

ROW 37 (194 : 198)

ROW 38 (198 : 204)

ROW 39 (204 : 206)

ROW 40 (206 : 211)

ROW 41 (211 : 215)

ROW 42 (216 : 220)

ROW 43 (220 : 228)

ROW 44 (228 : 231)

ROW 45 (232 : 236)

ROW 46 (236 : 241)

ROW 47 (241 : 243)

ROW 48 (243 : 249)

ROW 49 (249 : 254)

ROW 50 (254 : 256)

ROW 51 (256 : 261)

ROW 52 (261 : 263)

ROW 53 (264 : 267)

ROW 54 (267 : 274)

ROW 55 (275 : 285)

ROW 56 (286 : 289)

ROW 57 (290 : 298)

ROW 58 (298 : 304)

ROW 59 (305 : 309)

ROW 60 (310 : 314)

ROW 61 (314 : 321)

ROW 62 (322 : 328)

ROW 63 (328 : 332)

ROW 64 (332 : 335)

ROW 65 (336 : 344)

ROW 66 (344 : 349)

ROW 67 (349 : 355)

ROW 68 (356 : 363)

ROW 69 (363 : 368)

ROW 70 (369 : 369)

HEAT EXCHANGERS

PROGRAM REGISTERS NEEDED: 67

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 – 4)

ROW 2 (4 – 6)

ROW 3 (6 – 11)

ROW 4 (11 – 14)

ROW 5 (15 – 20)

ROW 6 (20 – 26)

ROW 7 (27 – 33)

ROW 8 (33 – 34)

ROW 9 (35 – 40)

ROW 10 (40 – 44)

ROW 11 (45 – 51)

ROW 12 (51 – 57)

ROW 13 (57 – 64)

ROW 14 (64 – 70)

ROW 15 (70 – 75)

ROW 16 (76 – 81)

ROW 17 (81 – 88)

ROW 18 (88 – 94)

ROW 19 (94 - 102)

ROW 20 (103 - 110)

ROW 21 (111 - 119)

ROW 22 (119 - 129)

ROW 23 (130 - 137)

ROW 24 (138 - 148)

ROW 25 (149 - 161)

ROW 26 (162 - 168)

ROW 27 (168 - 175)

ROW 28 (176 - 183)

ROW 29 (184 - 196)

ROW 30 (197 - 203)

ROW 31 (204 - 206)

ROW 32 (206 - 210)

ROW 33 (211 - 216)

ROW 34 (217 - 224)

ROW 35 (225 - 230)

ROW 36 (231 - 234)

HEAT EXCHANGERS:
PARALLEL FLOW SUBROUTINE
PROGRAM REGISTERS NEEDED: 8

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 – 6)

ROW 2 (7 – 19)

ROW 3 (19 – 25)

ROW 4 (26 – 36)

ROW 1 (1 - 6)

ROW 2 (7 - 18)

ROW 3 (19 - 28)

ROW 4 (28 - 37)

ROW 5 (38 - 49)

ROW 6 (50 - 59)

HEAT EXCHANGERS:
PARALLEL–COUNTER FLOW SUBR.
PROGRAM REGISTERS NEEDED: 12

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 – 4)

ROW 2 (5 – 17)

ROW 3 (18 – 26)

ROW 4 (26 – 34)

ROW 5 (35 – 47)

ROW 6 (48 – 60)

ROW 7 (61 – 62)

HEAT EXCHANGERS:
CROSS FLOW SUBROUTINE
PROGRAM REGISTERS NEEDED: 14

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 - 5)

ROW 2 (6 - 14)

ROW 3 (14 - 22)

ROW 4 (22 - 30)

ROW 5 (30 - 35)

ROW 6 (35 - 44)

ROW 7 (45 - 57)

ROW 8 (57 - 57)

DECIBEL ADDITION AND
SUBTRACTION
PROGRAM REGISTERS NEEDED: 8

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 – 4)

ROW 2 (5 – 9)

ROW 3 (10 – 16)

ROW 4 (17 – 26)

ROW 5 (26 – 26)

TEMPERATURE CONVERSION

PROGRAM REGISTERS NEEDED: 15

HEWLETT PACKARD
SOLUTION BOOK:
HEAT VENT & A/C

ROW 1 (1 - 5)

ROW 2 (5 - 7)

ROW 3 (8 - 12)

ROW 4 (12 - 16)

ROW 5 (17 - 23)

ROW 6 (23 - 28)

ROW 7 (29 - 33)

ROW 8 (33 - 37)

NOTES

NOTES

NOTES

NOTES

NOTES

NOTES

NOTES

HEWLETT
PACKARD

Rev. C

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

* Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

HEATING, VENTILATING & AIR CONDITIONING

OVERALL HEAT TRANSFER COEFFICIENT
INSULATION BREAK EVEN ANALYSIS
AIR FLOW IN CIRCULAT DUCTS
AIR DUCT CONVERSION
EQUATIONS OF STATE
BLACK BODY THERMAL RADIATION
PSYCHROMETRIC PROPERTIES
HEAT EXCHANGERS
DECIBEL ADDITION AND SUBTRACTION
TEMPERATURE CONVERSIONS

HEWLETT PACKARD

00041-90140 Rev. C 4/81 Printed in U.S.A.